# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=what4
pkgname=haskell-what4
pkgver=1.3
pkgrel=105
pkgdesc="Solver-agnostic symbolic values support for issuing queries"
url="https://github.com/GaloisInc/what4"
license=('BSD')
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-async' 'haskell-attoparsec' 'haskell-bimap' 'haskell-bifunctors'
         'haskell-bv-sized' 'haskell-concurrent-extra' 'haskell-config-value'
         'haskell-deriving-compat' 'haskell-data-binary-ieee754' 'haskell-fingertree'
         'haskell-hashable' 'haskell-hashtables' 'haskell-io-streams' 'haskell-lens' 'haskell-libbf'
         'haskell-panic' 'haskell-parameterized-utils' 'haskell-prettyprinter' 'haskell-scientific'
         'haskell-temporary' 'haskell-th-abstraction' 'haskell-th-lift' 'haskell-th-lift-instances'
         'haskell-unordered-containers' 'haskell-utf8-string' 'haskell-vector' 'haskell-versions'
         'haskell-zenc')
makedepends=('ghc' 'uusi' 'haskell-contravariant' 'haskell-hedgehog' 'haskell-lumberjack'
             'haskell-quickcheck' 'haskell-tasty' 'haskell-tasty-checklist'
             'haskell-tasty-expected-failure' 'haskell-tasty-hedgehog' 'haskell-tasty-hunit'
             'haskell-tasty-quickcheck' 'haskell-tasty-sugar')
checkdepends=('cvc4' 'yices' 'z3')
optdepends=('cvc4' 'yices' 'z3')
source=(https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz)
sha512sums=('b1abb534750815962fe77ac75edfe1d766557fbe4bb396da4f8e4a3564630886cccb84e1866babcd26ac9ce79286b921c9eb408b23f2a71446a0a8466ad492ee')

prepare() {
    cd $_hkgname-$pkgver
    gen-setup
    uusi -u tasty-sugar $_hkgname.cabal
}

build() {
    cd $_hkgname-$pkgver
    
    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
    cd $_hkgname-$pkgver
    runhaskell Setup test --show-details=direct
}

package() {
    cd $_hkgname-$pkgver
    
    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
    rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
